# Maintainer: Jan "heftig" Steffens <jan.steffens@gmail.com>
# Contributor: Caleb Cushing <xenoterracide@gmail.com>

pkgname=perl-ipc-system-simple
pkgver=1.30
pkgrel=3
pkgdesc="Run commands simply, with detailed diagnostics"
arch=('any')
url="https://search.cpan.org/dist/IPC-System-Simple"
license=('PerlArtistic' 'GPL')
depends=('perl')
options=('!emptydirs')
source=("https://search.cpan.org/CPAN/authors/id/J/JK/JKEENAN/IPC-System-Simple-$pkgver.tar.gz")
sha512sums=('82937df80ae38a9c21e43e256a21cdac1482d4a069eae282043d676ece06251a7139b752cd2b379a0f69ea9209ddd075607da86c6cb9b25f19cf90d91c28d72f')

build() {
  cd IPC-System-Simple-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd IPC-System-Simple-$pkgver
  make test
}

package() {
  cd IPC-System-Simple-$pkgver
  make DESTDIR="$pkgdir" install
}
